#幽匿观察者 找到目标
tag @n[tag=sculk_arrow.sculk_watcher.warden,distance=..1,type=warden] add warden
tag @s add watcher
$tag @n[nbt={UUID:$(owner)}] add owner
$tag @n[nbt={UUID:$(target)}] add target

#fail
execute if entity @n[tag=target,type=#dab:sculk_bow_no_find] run return run function dab:system/bow_arrow/sculk_bow/find_target.fail
execute unless entity @n[tag=target] run return run function dab:system/bow_arrow/sculk_bow/find_target.fail

execute as @n[tag=target] at @s run function dab:system/bow_arrow/sculk_bow/target_was_found
execute at @n[tag=watcher,type=item_display] as @n[tag=sculk_arrow.sculk_watcher.DM,distance=..1,type=marker] run function dab:system/bow_arrow/sculk_bow/record_targets with entity @n[tag=target]
playsound minecraft:entity.warden.tendril_clicks master @a ~ ~ ~ 1.2 1

data remove entity @n[tag=warden] listener.event
tag @s remove watcher
tag @n[tag=warden] remove warden
tag @n[tag=owner] remove owner
tag @n[tag=target] remove target